## 5.10 THE AIRFOIL OF ARBITRARY THICKNESS AND CAMBER

The analytical method of Sections 5.4 and 5.6 and the numerical method of Section 5.9 give remarkably accurate results for the thin, slightly cambered airfoils of conventional aircraft. On the other hand, the determination of the aerodynamic characteristics of thick, highly cambered, slotted surfaces, with single or multiple flaps and mutual interference effects among wings, fuselages, nacelles, and so forth, require, in general, the use of numerical methods such as the *source panel* representation described in Section 4.13. Since the method as described there applies only to nonlifting bodies, to treat lifting bodies it is necessary to introduce circulation, the strength of which is fixed by the Kutta condition. The accuracy of the method in practical flow problems is limited only by the skill of the designer in representing adequately the surface by source and vortex panels, by the number of simultaneous linear algebraic equations the computer can handle expeditiously, and by the accuracy to which the effects of viscosity and, at high flow speeds, compressibility can be included in the computation.

The following method is only one variation of the use of the panel method (Stevens et al., 1971); it involves representation of the airfoil by a closed polygon of vortex panels. The airfoil and wing problems can be solved by means of a vortex-panel distribution alone, but the calculation of fuselage and nacelle characteristics and their interference flows dictates the use of source and, possibly, doublet as well as vortex panels. Results of an analysis of a multiple-flap configuration using vortex panels only are given in Section 19.3.

The vortex-panel method introduced here has the feature that the circulation density on each panel varies linearly from one corner to the other and is continuous across the corner, as indicated in Fig. 5.23. The Kutta condition is easily incorporated in this formulation, and the numerical computation is stable unless a large number of panels is chosen on an airfoil with a cusped trailing edge. The panels, *m* in number, are assumed to be planar and are named in the clockwise direction, starting from the trailing edge. The "boundary points," selected on the surface of the airfoil, are the intersections of contiguous vortex panels; the condition that the airfoil be a streamline is met approximately by applying the condition of zero normal velocity component at "control points," specified as the midpoints of the panels.




Fig. 5.23. Replacement of an airfoil by vortex panels of linearly varying strength.

In the presence of a uniform flow  $V_{\infty}$  at an angle of attack  $\alpha$  and m vortex panels, the velocity potential at the *i*th control point  $(x_i, y_i)$  is, from Eqs. (2.67),

$$\phi(x_i, y_i) = V_{\infty}(x_i \cos \alpha + y_i \sin \alpha) - \sum_{j=1}^m \int_j \frac{\gamma(s_j)}{2\pi} \tan^{-1} \left(\frac{y_i - y_j}{x_i - x_j}\right) ds_j$$
 (5.43)

where

$$\gamma(s_j) = \gamma_j + (\gamma_{j+1} - \gamma_j) \frac{s_j}{S_j}$$
 (5.44)

As defined in Fig. 5.23,  $(x_j, y_j)$  represent coordinates of an arbitrary point on the jth panel of length  $S_j$ , which is at a distance  $s_j$  measured from the leading edge of the panel. The integration is performed along the entire panel from  $(X_j, Y_j)$  to  $(X_{j+1}, Y_{j+1})$ . Here, capital letters are used to denote the coordinates of boundary points. The (m + 1) values of  $\gamma_j$  at boundary points are unknown constants to be determined numerically.

The boundary condition requires that the velocity in the direction of the unit outward normal vector  $\mathbf{n}_i$  be vanishing at the *i*th control point, so that

$$\frac{\partial}{\partial n_i} \phi(x_i, y_i) = 0;$$
  $i = 1, 2, ..., m$ 

Carrying out the involved differentiation and integration in a manner similar to that shown in Section 4.13 for source panels, we obtain

$$\sum_{j=1}^{m} \left( C_{n \mid_{i_{j}}} \gamma_{j}' + C_{n \mid 2_{i_{j}}} \gamma_{j+1}' \right) = \sin(\theta_{i} - \alpha); \qquad i = 1, 2, ..., m$$
 (5.45)

in which  $\gamma' = \gamma/2\pi V_{\infty}$  is a dimensionless circulation density,  $\theta_i$  is the orientation angle of the *i*th panel measured from the *x* axis to the panel surface, and the coefficients are expressed by

$$C_{n1_{ij}} = 0.5DF + CG - C_{n2_{ij}}$$
  
 $C_{n2_{ij}} = D + 0.5QF/S_j - (AC + DE) G/S_j$ 

The constants shown in these and some later expressions are defined as

$$A = -(x_i - X_j)\cos\theta_j - (y_i - Y_j)\sin\theta_j$$

$$B = (x_i - X_j)^2 + (y_i - Y_j)^2$$

$$C = \sin(\theta_i - \theta_j)$$

$$D = \cos(\theta_i - \theta_j)$$

$$E = (x_i - X_j)\sin\theta_j - (y_i - Y_j)\cos\theta_j$$

$$F = \ln\left(1 + \frac{S_j^2 + 2AS_j}{B}\right)$$

$$G = \tan^{-1}\left(\frac{ES_j}{B + AS_j}\right)$$

$$P = (x_i - X_j)\sin(\theta_i - 2\theta_j) + (y_i - Y_j)\cos(\theta_i - 2\theta_j)$$

$$Q = (x_i - X_j)\cos(\theta_i - 2\theta_j) - (y_i - Y_j)\sin(\theta_i - 2\theta_j)$$

Note that these constants are functions of the coordinates of the *i*th control points, those of the boundary points of the *j*th vortex panel, and the orientation angles of both *i*th and *j*th panels. They can be computed for all possible values of *i* and *j* once the panel geometry is specified.

The expression in the parentheses on the left side of Eq. (5.45) represents the normal velocity at the *i*th control point induced by the linear distribution of vortices on the *j*th panel. The form of Eq. (5.45) corresponding to constant-strength vortex panels is shown in Chow (1979, Section 2.8). For i = j, the coefficients have simplified values

$$C_{n1n} = -1 \qquad \text{and} \qquad C_{n2n} = 1$$

which describe the self-induced normal velocity at the ith control point.

To ensure a smooth flow at the trailing edge, the Kutta condition (Eq. 5.3, which demands that the strength of the vorticity at the trailing edge be zero) is applied that, in the present notation, becomes

$$\gamma_1' + \gamma_{m+1}' = 0 5.46)$$

There are (m + 1) equations after combining Eqs. (5.45) and (5.46); they are sufficient to solve for the (m + 1) unknown  $\gamma'_j$  values. We may rewrite this system of simultaneous equations in a more convenient form:

$$\sum_{i=1}^{m+1} A_{n_{i,j}} \gamma'_{j} = RHS_{i}; \qquad i = 1, 2, ..., m+1$$
 (5.47)

in which, for i < m + 1:

$$A_{n_{i1}} = C_{n1_{i1}}$$

$$A_{n_{ij}} = C_{n1_{ij}} + C_{n2_{ij-1}}; j = 2, 3, ..., m$$

$$A_{n_{im+1}} = C_{n2_{im}}$$

$$RHS_{i} = \sin(\theta_{i} - \alpha)$$

and, for i = m + 1:

$$A_{n_{i1}} = A_{n_{im+1}} = 1$$
  
 $A_{n_{ij}} = 0; j = 2, 3, ..., m$   
RHS<sub>i</sub> = 0

Except for i = m + 1,  $A_{n_{ij}}$  may be called the normal-velocity influence coefficients representing the influences of  $\gamma'_i$  on the normal velocity at the *i*th control point.

After the determination of the unknown circulation densities, we now proceed to compute the velocity and pressure at the control points. At such a point, the velocity has only a tangential component at the panel surface because of the vanishing of the normal component there. Thus, if we let  $\mathbf{t}_i$  designate the unit tangential vector on the *i*th panel (see Fig. 5.23), the local dimensionless velocity defined as  $(\partial \phi/\partial t_i)/V_{\infty}$  can be computed, which has the expression

$$V_i = \cos(\theta_i - \alpha) + \sum_{j=1}^m \left( C_{t \mid_{i_j}} \gamma_j' + C_{t \mid_{i_j}} \gamma_{j+1}' \right); \qquad i = 1, 2, ..., m$$
 (5.48)

in which

$$C_{t1_{ij}} = 0.5CF - DG - C_{t2_{ij}}$$

$$C_{t2_{ij}} = C + 0.5PF/S_j + (AD - CE)G/S_j$$

$$C_{t1_{ij}} = C_{t2_{ij}} = \pi/2$$

The expression in the parentheses following the summation symbol has the physical meaning of the tangential velocity at the *i*th control point induced by the vortices distributed on the *j*th panel. To facilitate computer programming, Eq. (5.48) is further rewritten as

$$V_i = \cos(\theta_i - \alpha) + \sum_{j=1}^{m+1} A_{t_{ij}} \gamma_j'; \qquad i = 1, 2, ..., m$$
 (5.49)

where the tangential-velocity influence coefficients are defined as follows:

$$A_{t_{i_1}} = C_{t1_{i_1}}$$
  
 $A_{t_{i_j}} = C_{t1_{i_j}} + C_{t2_{i_{j-1}}}; j = 2, 3, ..., m$   
 $A_{t_{i_{m+1}}} = C_{t2_{i_m}}$ 

The pressure coefficient at the *i*th control point is, according to the definition of Eq. (3.19),

$$C_{pi} = 1 - V_i^2 (5.50)$$

The use of the vortex panel method just described is illustrated in the following example to compute the flow around a NACA 2412 airfoil flying at  $\alpha = 8^{\circ}$ . Figure 5.24




Fig. 5.24. Determination of panel boundary points on an airfoil. Hollow circles represent control points at centers of vortex panels.

shows a simple and yet reliable method for selecting boundary points on the airfoil. A circle centered at the midchord is drawn, which passes through both leading and trailing edges. When 12 panels are used in the present example, the circumference of the circle is divided into 12 arcs of equal length. Projection of the points on the circle gives 12 boundary points on the surface of the airfoil. The trailing edge is named twice as both the first and thirteenth boundary points, as shown in Fig. 5.24. A closed polygon of 12 panels is thus formed by connecting these boundary points. In this way, relatively short panels are automatically obtained in the leading- and trailing-edge regions, where changes in surface curvature are large.

Computations are performed on the computer by the use of a program written in FOR-TRAN language. This program can easily be modified to handle airfoils of arbitrary shape. The input of the program is a set of coordinates of the boundary points determined in the manner shown in Fig. 5.24 based on a smooth NACA 2412 airfoil, whose shape is obtained using a cubic-spline routine to fit the data points published by Abbott and von Doenhoff (1949), but with a minor modification to make the trailing edge close.

In this program, (X, Y) and (XB, YB) are used to represent coordinates of control and boundary points, respectively, and GAMA is the name used for  $\gamma'$ . Other names in the program are the same as those appearing in the analysis. Two subprograms, namely CRAMER and DETERM, are attached for solving a set of simultaneous algebraic equations using Cramer's rule.

```
PROGRAM AIRFOIL
C
                             C
С
          * THIS PROGRAM COMPUTES VELOCITY AND PRESSURE COEFFICIENT
C
             AROUND AN AIRFOIL, WHOSE CONTOUR IS APPROXIMATED BY M
С
             VORTEX PANELS OF LINEARLY VARYING STRENGTH.
С
С
          ******************
       PARAMETER (M = 12)
       DIMENSION XB(M+1), YB(M+1), X(M), Y(M), S(M), SINE(M), COSINE(M),
                 THETA (M), V(M), CP(M), GAMA(M), RHS(M), CN1(M,M),
                 CN2(M,M),CT1(M,M),CT2(M,M),AN(M+1,M+1),AT(M,M+1)
         SPECIFY COORDINATES (XB, YB) OF BOUNDARY POINTS ON AIRFOIL
С
C
         SURFACE. THE LAST POINT COINCIDES WITH THE FIRST.
       DATA XB
     * /1., .933, .750, .500, .250, .067,0.,.067,.250,.500,.750,.933,1./
       DATA YB
      /0.,-.005,-.017,-.033,-.042,-.033,0.,.045,.076,.072,.044,.013,0./
        MP1 = M+1
        PI = 4.0 * ATAN(1.0)
       ALPHA = 8. * PI/180.
С
         COORDINATES (X,Y) OF CONTROL POINT AND PANEL LENGTH S ARE
         COMPUTED FOR EACH OF THE VORTEX PANELS. RHS REPRESENTS
С
C
         THE RIGHT-HAND SIDE OF EQ. (5.47).
        DO I = 1, M
        IP1 = I + 1
        X(I) = 0.5*(XB(I)+XB(IP1))
        Y(I) = 0.5*(YB(I)+YB(IP1))
        S(I) = SQRT((XB(IP1)-XB(I))**2 + (YB(IP1)-YB(I))**2)
        THETA(I) = ATAN2((YB(IP1)-YB(I)), (XB(IP1)-XB(I)))
        SINE(I) = SIN(THETA(I))
        COSINE(I) = COS(THETA(I))
                = SIN( THETA(I)-ALPHA )
        RHS(I)
        END DO
        DO I = 1, M
        DO J = 1, M
        IF ( I .EQ. J ) THEN
        CN1(I,J) = -1.0
        CN2(I,J) = 1.0
        CT1(I,J) = 0.5*PI
        C\hat{T}2(I,J) = 0.5*PI
        ELSE
        A = -(X(I)-XB(J))*COSINE(J) - (Y(I)-YB(J))*SINE(J)
        B = (X(I) - XB(J)) **2 + (Y(I) - YB(J)) **2
        C = SIN(THETA(I)-THETA(J))
        D = COS(THETA(I)-THETA(J))
        E = (X(I)-XB(J))*SINE(J) - (Y(I)-YB(J))*COSINE(J)
        F = ALOG(1.0 + S(J)*(S(J)+2.*A)/B)
        G = ATAN2(E*S(J), B+A*S(J))
        P = (X(I)-XB(J)) * SIN(THETA(I)-2.*THETA(J))
          + (Y(I)-YB(J)) * COS( THETA(I)-2.*THETA(J) )
        Q = (X(I)-XB(J)) * COS(THETA(I)-2.*THETA(J))
          - (Y(I)-YB(J)) * SIN( THETA(I)-2.*THETA(J) )
        CN2(I,J) = D + .5*Q*F/S(J) - (A*C+D*E)*G/S(J)
        CN1(I,J) = .5*D*F + C*G - CN2(I,J)
        CT2(I,J) = C + .5*P*F/S(J) + (A*D-C*E)*G/S(J)
```

```
CT1(I,J) = .5*C*F - D*G - CT2(I,J)
        END IF
        END DO
        END DO
          COMPUTE INFLUENCE COEFFICIENTS IN EQS. (5.47) AND (5.49),
C
C
          RESPECTIVELY.
        DO I = 1, M
        AN(I,1)
                   = CN1(I,1)
        AN(I,MP1) = CN2(I,M)
        AT(I,1) = CT1(I,1)
        AT(I,MP1) = CT2(I,M)
          DO J = 2, M
          AN(I,J) = CN1(I,J) + CN2(I,J-1)
          AT(I,J) = CT1(I,J) + CT2(I,J-1)
           END DO
        END DO
        AN (MP1.1)
                    = 1.0
        AN(MP1, MP1) = 1.0
        DO J = 2, M
        AN(MP1,J) = 0.0
        END DO
        RHS(MP1) = 0.0
C
           SOLVE EQ. (5.47) FOR DIMENSIONLESS STRENGTHS GAMA USING
С
           CRAMER'S RULE.
                              THEN COMPUTE AND PRINT DIMENSIONLESS
C
           VELOCITY AND PRESSURE COEFFICIENT AT CONTROL POINTS.
        WRITE (6,6)
6
        FORMAT (1H1///11X, 1HI, 4X, 4HX(I), 4X, 4HY(I), 4X, 8HTHETA(I),
                    3x, 4HS(I), 3x, 7HGAMA(I), 3x, 4HV(I), 6x, 5HCP(I)/
     *
                    10X, 3H---, 3X, 4H----, 4X, 4H----, 4X, 8H------,
                    3x, 4H----, 3x, 7H-----, 3x, 4H----, 6x, 5H-----)
         CALL CRAMER ( AN, RHS, GAMA, MP1 )
        DO I = 1, M
         V(I) = COS(THETA(I)-ALPHA)
           DO J = 1, MP1
           V(I) = V(I) + AT(I,J)*GAMA(J)
           CP(I) = 1.0 - V(I)**2
           END DO
         WRITE(6,9) I, X(I), Y(I), THETA(I), S(I), GAMA(I), V(I), CP(I)
         END DO
9
         FORMAT (10X, I2, F8.4, F9.4, F10.4, F8.4, 2F9.4, F10.4)
         WRITE(6,10) MP1,GAMA(MP1)
10
         FORMAT (10X, I2, 35X, F9.4)
         STOP
         END
         SUBROUTINE CRAMER ( C, A, X, N )
           THIS SUBROUTINE SOLVES A SET OF ALGEBRAIC EQUATIONS
C
С
                C(I,J)*X(J) = A(I), I=1,2,---,N
           IT IS TAKEN FROM P.114 OF CHOW(1979)
C
         PARAMETER (M = 12)
         DIMENSION C(M+1,M+1), CC(M+1,M+1), A(M+1), X(M+1)
         DENOM = DETERM(C, N)
         DO K = 1, N
            DO I = 1, N
                DO J = 1, N
```

```
CC(I,J) = C(I,J)
             END DO
          END DO
          DO I = 1, N
          CC(I,K) = A(I)
          END DO
       X(K) = DETERM(CC, N) / DENOM
       END DO
       RETURN
       END
       FUNCTION DETERM( ARRAY, N )
          DETERM IS THE VALUE OF THE DETERMINANT OF AN N*N
          MATRIX CALLED ARRAY, COMPUTED BY THE TECHNIQUE
          OF PIVOTAL CONDENSATION. THIS FUNCTION IS TAKEN
          FROM PP.113-114 OF CHOW(1979)
       PARAMETER (M = 12)
       DIMENSION ARRAY (M+1, M+1), A (M+1, M+1)
       DO I = 1, N
       DO J = 1, N
       A(I,J) = ARRAY(I,J)
        END DO
       END DO
       L = 1
1
       K = L + 1
       DO I = K, N
        RATIO = A(I,L)/A(L,L)
          DO J = K, N
          A(I,J) = A(I,J) - A(L,J)*RATIO
          END DO
        END DO
        L = L + 1
        IF( L .LT. N ) GO TO 1
        DETERM = 1.
        DO L = 1, N
        DETERM = DETERM * A(L, L)
        END DO
        RETURN
        END
```

С С

С

C

I	X(I)	Y(I)	THETA(I)	S(I)	GAMA(I)	V(I)	CP(I)
1	0.9665	-0.0025	-3.0671	0.0672	-0.0823	-0.8585	0.2630
2	0.8415	-0.0110	-3.0761	0.1834	-0.1403	-0.8962	0.1969
3	0.6250	-0.0250	-3.0777	0.2505	-0.1422	-0.8890	0.2097
4	0.3750	-0.0375	-3.1056	0.2502	-0.1413	-0.8563	0.2667
5	0.1585	-0.0375	3.0925	0.1832	-0.1334	-0.7276	0.4707
6	0.0335	-0.0165	2.6839	0.0747	-0.0981	0.0840	0.9929
7	0.0335	0.0225	0.5914	0.0807	0.2170	1.6763	-1.8101
8	0.1585	0.0605	0.1678	0.1856	0.2785	1.5839	-1.5088
9	0.3750	0.0740	-0.0160	0.2500	0.2401	1.3905	-0.9334
10	0.6250	0.0580	-0.1115	0.2516	0.2098	1.2288	-0.5099
11	0.8415	0.0285	-0.1678	0.1856	0.1843	1.0811	-0.1688
12	0.9665	0.0065	-0.1916	0.0682	0.1578	0.9125	0.1674
13					0.0823		




Fig. 5.25. Calculated pressure distributions for three configurations of vortex panels.

The pressure coefficient distribution, based on the computer output for 12 panels, is plotted in Fig. 5.25. Plotted also in the same figure for comparison are the results for 48 and 120 panels, respectively, using the same computer program. It reveals that even with 12 panels, the pressure distribution is already close to the exact solution everywhere except in the region close to the leading edge. The large discrepancies in that region come from the fact that with 12 panels, the control points near the leading edge are far from the actual airfoil contour (see Fig. 5.25).

The total lift of the airfoil can be computed using the Kutta-Joukowski theorem of Section 4.8, in which the total circulation around the airfoil is the sum of contributions from all vortex panels. Such a computation is straightforward and is left as an exercise.

The panel method outlined here is considerably more cumbersome than exact methods, such as the conformal mapping technique for a single airfoil. However, the great power of the method emerges for flow calculations on multiple surfaces, such as airfoils with flaps and slots or cascades representing axial compressors or turbines and many other problems for which exact methods are, in general, not available (see Hess, 1971; Stevens et al., 1971). Corrections for compressibility and for viscous effects at high speeds or high angles of attack are discussed in the following chapters.

## 5.11 SUMMARY

The aerodynamic characteristics of airfoils of moderate camber and thickness have been derived on the hypothesis, verified by experimental results, that the shape of the mean camber line and the Kutta condition determine, to a good approximation, the aerodynamic characteristics. We show that for thin airfoils,  $dc_I/d\alpha = 2\pi$  and  $x_{AC} = \frac{1}{4}c$ . However,  $x_{AC}$  and  $c_{mac}$  are shown to be strongly influenced by the maximum mean camber and its location along the mean camber line; for an airfoil with flaps, these variables were inter-